AD-A173 845 APPLICATION OF A NOTCH DIGITAL FILTER TO ELIMINATION OF SINUSOIDAL DISTUR..(U) AERONAUTICAL RESEARCH LABS MELBOURNE (AUSTRALIA) R H PERRIN ET AL. APR 86 1/1

UNCLASSIFIED ARL-AERO-TM-382 F/G 12/1 NL



MICROCOPY RESOLUTION TEST CHART
NATIONAL BUREAU OF STANDARDS-1963-A

(12)

ARL-AERO-TM-382

AR-004-475

DEPARTMENT OF DEFENCE

DEFENCE SCIENCE AND TECHNOLOGY ORGANISATION

AERONAUTICAL RESEARCH LABORATORIES

MELBOURNE, VICTORIA

AD-A173 845

Aerodynamics Technical Memorandum 382

APPLICATION OF A NOTCH DIGITAL FILTER TO ELIMINATION OF SINUSOIDAL DISTURBANCES FROM HELICOPTER FLIGHT DATA

by

R.H. PERRIN AND R.A. FEIK

Approved for public release.

APRIL 1986

AR-004-475

DEPARTMENT OF DEFENCE
DEFENCE SCIENCE AND TECHNOLOGY ORGANISATION
AERONAUTICAL RESEARCH LABORATORIES

Aerodynamics Technical Memorandum 382

APPLICATION OF A NOTCH DIGITAL FILTER TO ELIMINATION OF SINUSOIDAL DISTURBANCES FROM HELICOPTER FLIGHT DATA

by

R.H. PERRIN AND R.A. FEIK

SUMMARY

Helicopter flight data, especially measurements of linear accelerations and angular velocities, are typically corrupted by sinusoidal deterministic disturbances, that are associated with the rotor frequency and its harmonics. In an effort to eliminate these disturbances without distorting the underlying trends, a new digital notch filter, developed under a research agreement between Aeronautical Research Laboratories (ARL) and the University of Newcastle, has been modified and implemented on the ARL ELXSI 6400 computer. In this memo, the filter is described and the frequency and transient response characteristics are summarised. Practical considerations arising out of application of single and cascaded filters to Sea King flight data are discussed, and the performance of the new filter is compared with that of comparable Butterworth filters.

POSTAL ADDRESS: Director, Aeronautical Research Laboratories,
P.O. Box 4331, Melbourne, Victoria, 3001, Australia.

CONTENTS

NOMENCLATURE

A	Attenuation (dB)
D	Numerator polynomial of filter equation
D_s	Denominator polynomial of filter equation
$E_o, E_1, ..E_n$	Numerator coefficients for digital Butterworth filter
F	Notch filter transfer function
$F_1, F_2 ... F_N$	Denominator coefficients for digital Butterworth filter
G	A factor such that the D.C. gain of the filter is unity.
N	Order of Butterworth filter
T	Sampling period
j	$(-1)^{1/2}$
f	Digital frequency (Hz)
f_c	Digital cut-off frequency of Butterworth filter (Hz)
q	Delay operator
s(t)	Disturbance value at time t
t	Time (s)

$y(t)$	True signal at time t
$y_f(t)$	Filtered signal at time t
$y_m(t)$	Measured signal at time t
α	Parameter determining sharpness of filter notch
Ω	Analogue frequency (rad/s)
Ω_c	Analogue cut-off frequency (rad/s)
τ	Phase or time delay of Butterworth filter (s)
ω	Digital frequency (rad/s)
ω_c	Digital cut-off frequency (rad/s)
ω_o	Digital cut-out frequency (rad/s)

1. INTRODUCTION

In 1979, flight trials of the Sea King Mk 50 helicopter were performed in order to validate a mathematical model of the helicopter, developed in the Aircraft Behaviour Studies - Rotary Wing (ABS-RW) Group, at Aeronautical Research Laboratories (ARL) In an attempt to remove unwanted noise from the data records, the recorded signals were pre-processed using analogue Butterworth filters. However, considerable noise still existed on the data obtained. To remove this noise, one of two digital Butterworth filters was used [1].

In a separate study, under a research agreement on flight path reconstruction, workers at the University of Newcastle investigated methods for the rejection of deterministic sinusoidal disturbances. An optimal Kalman filtering approach was found to be impractical [2]. However, additional research showed that the optimal filter could be approximated, to a high degree of accuracy, by a digital filter which was both simple and practical to use [3, 4, 5].

A simple program presented in Ref. 3 has been substantially modified to adapt it for use as part of the flight reduction procedures in the ABS-RW group.

In Section 2, the theoretical derivation of the filter is summarised briefly, while Section 3 discusses practical issues involved with the application the filter to flight data. Finally, in Section 4, the performance of the new filter is compared with that of a comparable Butterworth filter.

2. DERIVATION OF THE DIGITAL NOTCH FILTER

The aim was to design a filter which would remove a sinusoidal disturbance from a measured signal, $y_m(t)$, to produce a filtered signal, $y_f(t)$. In the limit, it was desired that the filtered signal should approach the original signal y(t) (see Fig.1).

The detailed design and the underlying theory of this filter can be found in Refs. 2 to 5. Excerpts from the above references are reproduced below.

It can be shown that a sinusoidal disturbance, s, of frequency ω_o, obeys the following difference equation (Ref. 2, p2.2).

$$D(q^{-1})s(t) = 0 \tag{1}$$

where q^{-1} is the delay operator and

$$D(q^{-1}) = 1 - 2\cos(\omega_o T)q^{-1} + q^{-2} \tag{2}$$

Let the filter transfer function, F, be of the form

$$F(q^{-1}) = \frac{D(q^{-1})}{D_s(q^{-1})} \tag{3}$$

with

$$D_s(q^{-1}) = 1-\alpha.2\cos(\omega_o T)q^{-1}+\alpha^2 q^{-2} \tag{4}$$

where α is a parameter such that $0 \leq \alpha \leq 1$ and ω_o is the frequency being filtered.

If the measured signal is the sum of the original signal, y(t), and the disturbance, s(t), we have

$$y_m(t) = y(t)+s(t) \tag{5}$$

From Ref. 1, page 3, we have the following relationship between the filtered and original signals

$$D_s(q^{-1}) \{y_f(t)-y(t)\} = \{D(q^{-1})-D_s(q^{-1})\} \; y(t) \tag{6}$$

Hence, if D_s can be chosen to be close to D, while still retaining stability in the system as a whole, $y_f(t)-y(t)$ will, in the limit, be small.

Using Equations (4) and (6), this similarity between D_s and D can be achieved by selecting a value for α close to 1.

The final form of the filter can be expressed as

$$F(q^{-1}) = G \frac{D(q^{-1})}{D_s(q^{-1})} \tag{7}$$

$$= G \frac{(1-2\cos(\omega_o T)q^{-1}+q^{-2})}{(1-\alpha 2\cos(\omega_o T)q^{-1}+\alpha^2 q^{-2})} \tag{8}$$

where G is a scaling factor to ensure that the D.C. gain of the filter is unity and is given by

$$G = \frac{1-\alpha\ 2\cos(\omega_o T)+\alpha^2}{1-2\cos(\omega_o T)+1} \tag{9}$$

From Equation (8), it can be seen that if the disturbance frequency, ω, is equal to ω_o, exact cancellation of the disturbance will occur regardless of the value of α being used. However, for ω not equal to ω_o, the gain of the filter is close to one, provided α is close to one, so that the signal passes with minimal alteration. Thus the filter resembles a notch at ω_o with α a measure of the notch width.

3. USE OF THE NOTCH FILTER

3.1 Discussion of Program

The original program presented in Ref.3 has been modified and integrated into an existing helicopter data analysis program [6]. Improvements were necessary to cope with the cascading of a large number of filters and to obtain estimates of initial conditions. The equations used in the program are contained in Section 2. A complete listing of the ARL program can be found in Appendix A, while Fig.2 shows the flow of the program during execution.

Execution of the program requires the user to type two responses (the name of the file containing the data to be filtered and whether the filter characteristics are required in the output file), with the remainder of the execution being controlled by an input file called 'INIT.DAT'. The contents of 'INIT.DAT' are explained in the header of the main program.

3.2 Selection of Parameter Values α, ω_o

3.2.1 Background

Implementation of the digital notch filter is conceptually simple as only two parameters, α and ω_o, need to be selected. Use of the filter in the theoretical case, described in Section 2, where ω is known exactly, simply requires the substitution of $\alpha = 1$ and $\omega_o = \omega$ into the equations. This yields a filter

which exactly cancels out the noise signal. However, in applying the filter to real data, a number of practical considerations need to be addressed as described below.

3.2.2 Choice of the Value for α

The major problem is that the noise frequencies in a record of data will very rarely, if ever, be known exactly, resulting in an estimate for ω_o being used in the filter.

Thus, if a value of α close to 1 is used, the resulting narrow notch generated by the filter may result in the noise signal being missed entirely. Figures 3, 4, and 5, show the Bode plots for the filter for varying values of α, demonstrating the effect of α on notch width. It can be seen that as α approaches 1, the notch width decreases, and distortion to frequencies other than the cut-out frequency, ω_o, decreases.

In addition to its effect on notch width, the value of α also affects the transient response characteristics of the filter. As shown by Ref. 2, Fig.2, and reproduced here as Fig.6, as the value of α tends to 1, the roots of the denominator polynomial, $D_s(q^{-1})$, approach the boundary of the unit circle. This results in lower filter damping. This is illustrated in Fig.7, which shows responses of the filter to a step input for different values of α. These transients can be a problem if the initial conditions are offset, or if rapid changes occur in the signal. In the former case, the filter can be started early, if sufficient signal is available, and the oscillations allowed to decay before the region of interest is reached. However, excitation of transient oscillations during the record are not as simple to remove. Figure 7(b) shows the response of the filter to the same step input as that used in Fig.7(a), but with the filter run over the data twice, in forward and reverse time direction, a technnique used when cascading filters, as discussed below. The figure shows that the effects of ringing persist longer at the higher values of α. The possible presence of such oscillations

should be ascertained when examining a filtered signal, and allowances made if they do exist.

The practical use of this filter involves reaching a compromise in choosing the value of α. While sharpening the notch lessens the effect of the filter on other frequencies, it decreases the chance of filtering the signal if a slightly erroneous value of ω_o is selected, and also increases the chance of transient ringing. A method used to partially offset the effect of notch width is to use a series of cascaded filters instead of a single filter. By cascading the filters, higher values of α are able to be used, resulting in less distortion of frequencies other than those being filtered. However, the probability of 'ringing' also increases.

3.2.3 Cascaded Filter Spacing

Associated with the cascading of a series of filters, the spacing of the filters, dF, becomes an important parameter. A large value of dF (depending on α) can allow intermediate frequencies to be passed, degrading the quality of the filtered signal. Similarly, if the spacing is too small, then computer time is being wasted by repeatedly filtering the same data. Figures 8, 9, 10, and 11 show the response of the filter when cascaded between two nominal frequencies, 3.3 and 3.7 Hz.

3.3 Initial Conditions

Errors in the initial conditions lead to an initial transient response in the filtered data as explained in the previous section. In order to estimate the initial conditions, a subroutine 'INITIAL' can be called. In 'INITIAL', the filter is initialised by setting the first two filtered points equal to the initial data point, and the data is then run through the filter allowing any transients to damp out. Following this, the final values are used to re-initialise the filter which is then run in reverse to give the required initial conditions.

In practice, the resulting initial conditions depend on the value of ω_0 chosen for the filter: thus it is important to choose ω_0 to match the disturbance frequency so as to minimise any residual transients. If a wide range of frequencies is present, then the filter can be re-initialised at suitable intervals if necessary.

3.4 Phase Shift

As shown is Figures 3, 4, and 5, a phase shift can result from the use of a single filter. These figures show that the phase shift is a function of the value of α used. As α tends to 1, the region affected by the shift in phase contracts to the vicinity of the notch frequency, ω_0, and phase shifts can be neglected at other frequencies. As outlined in Section 3.2.2, the use of a single filter is not a practical method for the application under consideration, in contrast to cascading. By running the cascaded filters in opposite directions, any phase shifts introduced by the filters tend to cancel each other. Comparisons of phase shift diagrams of Figs. 3, 4, and 5 with those of Figs. 8, 9, 10, and 11 show the reduction in the net phase shift of the signal as a result of the cascaded filtering process.

3.5 Example using Spectral Information

Because of the ability of the filter to remove specific noise frequencies from the signal record, use can be made of spectral analysis of the signal to identify the frequencies present. For example, Fig.12 shows a vertical acceleration signal and Fig.13 the noise power spectrum of the unfiltered signal. Examination of the noise power spectrum shows that there exists considerable noise with frequencies in the region 16 to 19 Hz and in the 1 to 5 Hz region. To illustrate the selective filtering capability, the 16 to 19 Hz noise was removed using filters with $\alpha = 0.99$ and a spacing of 0.05 Hz. The resulting output signal is

shown in Fig.14. The power spectrum (Fig.15) of the filtered signal shows the removal of noise with frequencies in the 16 to 19 Hz region, while other frequencies are unaffected.

4. COMPARISON BETWEEN NOTCH AND BUTTERWORTH FILTERS

4.1 Overview of the Butterworth Filter

The purpose of developing the new notch filter was to produce a filter which performed better, when applied to actual flight data, than the Butterworth filters currently used. The characteristics of these Butterworth filters are shown in Ref.5, page 8 and are reproduced below as Table 1.

The Bode plots of the two filters, defined in Table 1, are shown as Fig.16a (the high cut-off frequency filter) and Fig.16b (the low cut-off frequency filter). These two figures show that signal frequencies appreciably higher than the cut-off frequency (e.g. to 8 Hz and higher in Fig.16a) can still be passed, although in attenuated form. This characteristic means that if all signal frequencies above a certain frequency needed to be removed from the data record, then the cut-off frequency, f_c, has to be set at some point well below that frequency. This may result in the distortion of the base signal.

Phase shifts resulting from the use of Butterworth filters are greater than those incurred by using cascaded notch filters. However, this can be compensated for by either of two methods (i) shift the data by a factor, τ (see Ref.5) or (ii) filter the signal from both ends of the data record to achieve phase cancellation.

4.2 Bode Plot Comparisons

When comparing the Butterworth and the new notch filter, it should be remembered that the notch filter is characterised by a cut-out frequency while the Butterworth is characterised by a cut-

off frequency. The notch filter can be made to emulate the low-pass Butterworth filter under consideration by cascading filters to remove all signal frequencies above a specific value.

The cascaded notch filter used for the comparison has α=0.99 with filters between 4 and 16 Hz at a spacing of 0.05 Hz. (Fig.17 shows a Bode plot of this cascaded filter). As indicated by Fig.17, the filter completely removes the signals beyond 4 Hz while having little effect on the amplitude of lower signal frequencies. Additionally, the phase shift of the filter is very small and can be ignored. Figure 18 superimposes the Bode plots of the two filters and shows that the notch filter has a more square drop-off and smaller phase shift.

4.3 Comparison using Actual Flight Data

The performances of the Butterworth and notch filters can be compared by use of the filters on actual flight data. Two data records are used for the comparison, the first being the vertical acceleration signal and the second, the pitch rate signal, both from flight number 17014. The acceleration signal contained sharp variations, while the pitch rate signal was slowly varying. These records represent the two extremes in signal type likely to be encountered in the flight measurements.

Firstly, in filtering the vertical acceleration signal, the notch filter used was divided up into two parts. The first part comprised a series of cascaded filters between 6 and 30 Hz witn α=0.975 and dF=0.1 Hz (designed to remove all high frequency noise), while the second part was a series of filters between 3 and 6 Hz with α =0.99 and dF=0.05 Hz. The second bank of filters needed a higher value of α to minimise distortion to the base signal (which appears to have some frequency components in the region of 2.5 to 3 Hz), while still filtering out the 3.5 Hz rotor frequency noise. The Butterworth filters used were those defined by Table 1. Figure 19 displays the comparison between the two filter types.

The distortion of the vertical acceleration signal as a result of using the low cut-off frequency Butterworth filter is clearly evident and would indicate an inadequate filter choice for this type of signal, where sharp variations are present. Use of the high cut-off frequency Butterworth filter appears to be a better alternative, as its output closely follows the trends of the original signal. *Noticeable* in this signal is the presence of residual noise with approximately 3.5 Hz frequency, indicating that a considerable component of the original noise signal is still being passed.

TABLE 1

Specifications of Standard Digital Butterworth Lowpass Filters

Quantity	Filter 1 (low cut-off frequency)	Filter 2 (high cut-off frequency)
T (s)	1/60	1/60
N	5	5
f (Hz)	3·5	4·0
A (dB)	50·0	3·0
f_c (Hz)	1·12	4·0
Ω_c (rad/s)	7·0331292892882D+00	2·5518903306118D+01
τ (s)	0·459	0·127
E_0, F_0	1, 1·7478224133541D+06	1, 4·5580802280576D+03
E_1, F_1	5, −8·0769814534886D+06	5, −1·6621499551730D+04
E_2, F_2	10, 1·4953389527402D+07	10, 2·4877628218574D+04
E_3, F_3	10, −1·3862462739192D+07	10, −1·8991475064559D+04
E_4, F_4	5, 6·4345340695383D+06	5, 7·3691389324960D+03
E_5, F_5	1, −1·1962698176138D+06	1, −1·1598727628397D+03

Use of the notch filter as an alternative to the high cut-off frequency Butterworth filter seems a more favourable alternative, as it has the same overall quality in the filtered signal as the Butterworth filter, with the added advantage of having the 3.5 Hz noise removed from the data record. However, some residual oscillations, in the region of sharp variations in the original signal, still remain. These are probably due to the filter transient characteristics and depend on the choice of α, as discussed in Section 3.2.2. More commonly, the notch filter would be used on slowly varying signals of which the pitch rate signal from this flight appears to be typical. Figure 19(b) shows the comparison between the unfiltered pitch rate signal, the notch filtered signal, and the two Butterworth filtered signals. The high cut-off frequency Butterworth filter would be unsuitable for this record type as there is still considerable noise remaining in the filtered data record. However, the low cut-off frequency Butterworth filter, although it has removed the noise from the data, it seems to have introduced some low frequency oscillations into the filtered data which are not apparent in the original data.

Figure 19 (b) indicates that the notch filter would be an acceptable filter choice, as the noise signal has been removed, while the base signal remains uncorrupted. The filter used had α=0.90, dF=0.10 Hz, with notches cascaded between 3 and 30 Hz. This figure shows that if the signal is slowly varying, then a cascade of notch filters with values of α in the region of 0.90 to 0.95 can be used with adequate results. Accompanying the low value of α, large filter spacings (dF $\approx$ 0.10 Hz), and hence fewer filters, can be used to filter a record. This results in some improvement in the CPU time used when compared to cascaded filters with higher values of α for the same data record.

4.4 Comparison of Execution Times for the Two Filter Types

As implemented here, the notch filter processes the data many times as it filters, while the Butterworth filter processes

the data only once. Hence, despite its simplicity, the run times of the notch filter can be significantly longer than those of the Butterworth filter. In the application described in Section 4.3, involving a total of 330 cascaded filters, the time taken for the high cut-off frequency Butterworth filter was less by a factor of 4.25 than the notch filter for the same range of filtering. (1.3 CPU seconds compared with 5.5 CPU seconds).

The run time of the notch filter is proportional to the number of filters in the cascade. For small variations of the noise frequency, for example if the noise is confined to, say 17 to 18 Hz, then the notch filter would take considerably less time than the corresponding Butterworth filter.

5. CONCLUDING REMARKS

A simple, near optimal digital notch filter designed under a research agreement with the University of Newcastle has been adopted for use at ARL for the removal of unwanted noise from helicopter flight data. The ARL implementation of the filter allows it to be used conveniently in either of two modes of operation. If the noise power spectrum indicates the presence of noise at specific frequencies, then a single filter or cascade of filters can be chosen to remove particular frequencies of concern. Guidance has been provided for the choice of notch width and filter spacing.

If the noise spectrum covers a wide range of frequencies, or else if a power spectrum is not immediately available, then a block filtering technique can be used to remove all frequencies above a chosen value (low pass filter). A relatively wide notch and wide filter spacing can be used to reduce computing time. A combination of the above two operating modes can be used if required.

Comparisons have been made with two Butterworth filters currently used to filter the flight data, and an indication of the

performance of the notch filter obtained. The comparisons showed that the new filter has a superior frequency response, both in magnitude and phase shift, when compared to the existing Butterworth filters. However, execution times can be several times greater if a large number of filters are cascaded together. Also, the transient response characteristics of the filter can cause problems with rapidly varying records if the notch width is very narrow. Further study could be aimed at circumventing this potential problem area.

In general, the new filter is relatively easy to use and to program for a computer. It provides a useful addition to the suite of programs used in the ABS-RW group for helicopter flight data analysis. Its simplicity makes it straight forward to use in other applications.

ACKNOWLEDGEMENT

The authors gratefully acknowledge Dr N. E. Gilbert for advice.

REFERENCES

1. Gilbert, N.E. and Fleming, J.A. - "Digital Filtering of Helicopter Flight Data" ARL Aero Note 406, January 1982.

2. Lozano-Leal, R., Goodwin, G.C. and Evans, R.J. - "Rejection of Sinusoidal Disturbances from Helicopter Flight Data". University of Newcastle Technical Report EE 8420, June 1984.

3. Lozano-Leal, R., Goodwin, G.C. and Evans R.J. -"Users Guide for Filter to Reject Sinusoidal Disturbances". University of Newcastle Technical Report EE 8434, October 1984.

4. Goodwin, G.C. and Lozano-Leal, R. - "A Digital Filter for Sinusoidal Disturbance Rejection With Application to Helicopter Flight Data Estimation". University of Newcastle, New South Wales, Australia.

5. Goodwin, G.C. and Evans, R.J. - "Application of Fast Fourier Transforms to Sinusoidal Disturbance Rejection and its Relationship to Kalman Filtering". University of Newcastle Technical Report EE 8433, October 1984.

6. Gilbert, N.E. - "Data Reduction Procedures for Sea King Helicopter Flight Trials" ARL Aero Technical Memo 338, May 1982.

APPENDIX A : Listing of the filtering program

```
      PROGRAM FILTER
C*****************************************************************
C
C         This program cancels sinusoidal disturbances out of an
C     arbitary signal in a given frequency range. The filter is of
C     the form:
C
C     G*(1-2cos wTq**-1 + q**-2)/(1-ALPHA.2cos wTq**-1+ALPHA**2.q**-2)
C
C         Where :
C
C     T     : Sampling period.
C     w     : 2 * PI * FRE
C     FRE   : Lowest frequency component in the sinusoidal disturbance
C     ALPHA: (0 - 1) Filter's sensitivity to inaccuracies in the
C             value of FRE.
C             Typically < 0.95 - 0.99>
C     G     : A coefficient such that the filter's gain is 1.0
C
C
C         Other variables in the program are :
C
C     N     : Length of the data
C     NF    : Number of filters in cascade
C     FRE   : Frequency used in the design of the first filter in Hz
C     FREXP: The expected value of the noise's frequency in Hertz
C             Used in finding the initial conditions
C     DF    : Increment in frequency from one filter to the next
C             Typically  < 0.1Hz - 1.0Hz >
C     NPAS : Number of passes through the data used to find
C             the initial condition for the filter output. This is
C             done for each one of the filters.
C
C         Files used in the program :
C
C     "INFILE"     : Contains the UNfiltered signal.
C     CLEAN.DAT    : Contains the filtered signal.
C     INIT.DAT     : Contains the data for the different variables
C                    used in the program.
C
C
C         LAST UPDATE  :  24th June  1985
C
C*******************************************************************
      IMPLICIT DOUBLE PRECISION ( A-H, O-Z)
      DATA PI/3.141592654/
      DOUBLE PRECISION Y(3), VER(6000), AUX(6000)
      CHARACTER*500 INFILE
      CHARACTER*1 ANS
1     WRITE (5,2)
2     FORMAT (9(/),' D I G I T A L   N O T C H   F I L T E R I N G',/
     1       ' *******************************************************'//,
     2 ' What is the name of the input file?'//)
      READ (5,3) LENGTH, INFILE
3     FORMAT ( Q, A500 )
      OPEN (UNIT=20,FILE=INFILE(1:LENGTH),STATUS='OLD',ERR=1)
      OPEN (UNIT=30,FILE='INIT.DAT',STATUS='OLD')
      OPEN (UNIT=2,FILE='CLEAN.DAT',STATUS='UNKNOWN')
C
      READ (30,*) FRE, FREXP, ALPHA, T, DF
      READ (30,*) NPAS, N, NF
C
      WRITE (5,5)
5     FORMAT (//' Output the filters characteristics ?'$)
      READ (5,6) ANS
6     FORMAT (A1)
      IF ( ANS .EQ. 'Y' .OR. ANS .EQ. 'y' ) THEN
         WRITE (2,7) INFILE (1:LENGTH)
7        FORMAT (' Input filename :',A20)
         WRITE (2,8) FRE, FREXP, ALPHA, T, DF, NPAS, N, NF
8        FORMAT (' ',' FRE   =',F15.8,/,' FREXP=',F15.8,/,
     1     ' ALPHA=',F15.8,/,' T     =',F15.8,/,
     2     ' DF    =',F15.8,/,' NPAS =',I5,/,' N     =',I5,/,' NF    =',I5)
      END IF
C
C
C  Read in the data to be filtered from tape 20
C
      DO 10 I = 1,N
10       READ (20,2030) Z, VER(I)
C
C  Check to see whether the data needs to have the
C correct initial conditions to be found.
C
      IF ( NPAS .GT. 0 ) THEN
         CALL INITIAL (FREXP,ALPHA,T,NPAS,N,VER,Y)
         Y(3)  =  Y(1)
         Y(2)  =  Y(2)
      ELSE
         Y(3)  =  VER(1)
         Y(2)  =  VER(1)
      ENDIF
```

```
C
C  This section represents the main loop of the program.
C
      FRE  =  FRE - DF
      DO 2000 II = 1, NF
         FRE  =  FRE + DF
C
C        WRITE (5,20)FRE
C 20     FORMAT ('Filter frequency :',F10.5)
C
C  Calculate the value of 2.Cos(Wk.T) and the filter's D.C. gain 'G'
C
         A  =  2.0 * DCOS(2.0*PI*FRE*T)
         G  =  ( 1.0 - ALPHA*A + ALPHA*ALPHA ) / ( 2.0 - A )
C
C  Perform the filtering task for the remainder of the
C data points
C
         DO 100 I = 3, N
            E  =  VER(I) - A*VER(I-1) + VER(I-2)
            Y(1)  =  ALPHA*A*Y(2) - ALPHA*ALPHA*Y(3) + E*G
C
C  Over-write the old value of Y(3) to the data
C
            VER(I-2)  =  Y(3)
C
C  Shift the data points Y(3) and Y(2) to the next
C data point
C
            Y(3)  =  Y(2)
            Y(2)  =  Y(1)
C
 100     CONTINUE
C
C  Write out the two filtered end points to the variable
C array VER
C
         VER(N-1)  =  Y(3)
         VER(N)  =  Y(2)
C
C
C  Modify the values for Y(3) and Y(2) so that they contain
C the correct values for the reverse sweep of the data
C
         Y(2)  =  Y(3)
         Y(3)  =  Y(1)
C
C  Load into the array AUX the reversed contents of the
C array VER to enable the backwards pass through the data
C
        DO 200 I = 1, N
 200       AUX(I)  =  VER(N+1-I)
C
C  Copy the contents of the array AUX into the array VER
C
        DO 300 I = 1, N
 300       VER(I)  = AUX (I)
C
C
 2000 CONTINUE
C
C
C
C  Check to see wHether there exists a condition that
C would result in the reversal of the data and if so
C ensure that the data is written out in the correct form
C
      IREV = 2 * INT ( FLOAT (NF) / 2.0 )
      IF ( IREV .EQ. NF ) GOTO 2015
C
      DO 2005 JJ = 1,N
 2005    AUX (JJ)  =  VER ( N + 1 - JJ )
      DO 2010 JJ = 1,N
 2010    VER ( JJ ) = AUX ( JJ )
C
C
 2015 CONTINUE
C
C
      DO 2020 I = 1,N
         TIME  =  T * FLOAT (I-1)
 2020    WRITE (2,2030) TIME, VER(I)
C
C  Call routine which calculates the CPU time used
C in the running of the program.
C
      CALL CPTIME
C
 2030   FORMAT(2X,2(F15.7,2X))
      END
```

```
      SUBROUTINE INITIAL (FRE,ALPHA,T,NPAS,N,VAL,Y)
C**********************************************************************
C
C         The purpose of this subroutine is to find the initial
C      conditions for the data to enable a single pass filtering
C      to be performed.
C
C         Last update  :  20th June 1985
C
C**********************************************************************
      IMPLICIT DOUBLE PRECISION ( A-H, O-Z )
      DIMENSION  Y(3), VAL(6000), AUX(6000)
      DATA PI /3.141592654/
C
C        The following is used to ascertain whether NPAS is
C      odd or even since for correct operation of this subroutine
C      ******    NPAS MUST BE ODD    ******
C
      IREV  =  2 * INT ( FLOAT(NPAS)/2.0 )
      IF ( IREV .EQ. NPAS ) THEN
         NPAS  =  NPAS + 1
      ENDIF
C
C         Initialise the values for Y(3) and Y(2) to the first value
C      in the data to be filtered
C
      Y(2)  =  VAL(1)
      Y(3)  =  VAL(1)
C
C         Calculate the value for 2.Cos(Wk.T) and the
C      filter's D.C. gain value 'G'
C
      A  =  2.0 * DCOS(2.0*PI*FRE*T)
      G  =  ( 1.0 - ALPHA*A + ALPHA*ALPHA ) / ( 2.0 - A )
C
C         This section represents the main loop of the subroutine
C
 10   CONTINUE
C
      DO 100 I = 3, N
         E  =  VAL(I) - A*VAL(I-1) + VAL(I-2)
         Y(1)  =  ALPHA*A*Y(2) - ALPHA*ALPHA*Y(3) + E*G
C
C         Shift the data points Y(2) and Y(3) for the next calculation
C
         Y(3)  =  Y(2)
         Y(2)  =  Y(1)
C
 100   CONTINUE
C
C         Load into array AUX the reversed contents of the array
C       VAL to enable the backwards pass to be made
C
       DO 200 I = 1, N
 200      AUX(I)  =  VAL(N+1-I)
C
C       Copy the contents of the array AUX into the array VAL
C
       DO 300 I = 1, N
 300      VAL(I)  =  AUX(I)
C
C         Decrease the value of NPAS by 1
C
       NPAS  =  NPAS - 1
       Y(2)  =  Y(3)
       Y(3)  =  Y(1)
       IF ( NPAS .GE. 0 ) GOTO 10
C
       RETURN
       END
```



Figure 1 General Overview of the Filtering Process



Figure 2 Flow Diagram for the Notch Filter Program



Figure 3 **Frequency Response of the Notch Filter for ALPHA = 0.900**



Figure 4 **Frequency Response of the Notch Filter for ALPHA = 0.950**



Figure 5 **Frequency Response of the Notch Filter for ALPHA = 0.999**



Figure 6 **Configuration of the Roots of the Notch Filter**



Figure 7(a) **Comparison of the Step Response of the Notch Filter for Various Values for ALPHA**



Figure 7(b) **Comparison of the Transient Responses of the Notch Filter to a Step Input for Various Values of ALPHA**



Figure 8(a) **Frequency Response of Cascaded Notch Filter for ALPHA=0.950 and dF=0.20 Hz**



Figure 8(b) Frequency Response of Cascaded Notch Filter for ALPHA=0.950 and dF=0.10 Hz



Figure 9 Frequency Response of Cascaded Notch Filter for ALPHA=0.975 and dF=0.10 Hz



Figure 10(a) Frequency Response of Cascaded Notch Filter for ALPHA=0.990 and dF=0.10 Hz



Figure 10(b) Frequency Response of Cascaded Notch Filter for ALPHA=0.990 and dF=0.050 Hz





Figure 11(a) Frequency Response of Cascaded Notch Filter for ALPHA=0.999 and dF=0.050 Hz



Figure 11(b) Frequency Response of Cascaded Notch Filter for ALPHA=0.999 and dF=0.005 Hz



Figure 12 **Original Unfiltered Vertical Acceleration Signal**



Figure 13 **Noise Power Spectrum for the Unfiltered Vertical Acceleration Signal**



Figure 14 Vertical Accleleration Signal after Notch Filtering with Filter between 16 and 19 Hz with ALPHA = 0.99



Figure 15 **Power Spectrum of the Vertical Acceleration Signal after Notch Filtering with Filters between 16 and 19 Hz with ALPHA = 0.09**



Figure 16(a) Frequency Response of the High Cut-Off Frequency Butterworth Filter





Figure 16(b) Frequency Response of the Low Cut-Off Frequency Butterworth Filter



Figure 17 Frequency Response of the Notch Filter with Cascaded Filters between 4 and 16 Hz



Figure 18 **Superimposed Frequency Responses for the Notch and High Cut-Off Frequency Butterworth Filters**



Figure 19(a) **Comparison between the Unfiltered and Filtered Vertical Acceleration Signals using the Notch and Various Butterworth Filters**



Figure 19(b) Comparison between the Unfiltered and Filtered Pitch Rate Signals using the Notch and Various Butterworth Filters

DISTRIBUTION

AUSTRALIA

Department of Defence

Defence Central

Chief Defence Scientist)
Deputy Chief Defence Scientist)
Superintendent, Science and Program Administration)1 copy
Controller, External Relations, Projects and
Analytical Studies)
Defence Science Adviser (UK) (Doc Data Sheet Only)
Counsellor, Defence Science (USA) (Doc Data Sheet Only)
Defence Central Library
Document Exchange Centre, DISB (18 copies)
Joint Intelligence Organisation
Librarian H Block, Victoria Barracks, Melbourne

Aeronautical Research Laboratories

Director
Library
Superintendent - Aerodynamics
Divisional File - Aerodynamics
N.E. Gilbert
N. Matheson
M.J. Williams
A.A. Arney
D.A. Secomb
C.A. Martin
A.J. Farrell
C. Sutton
K. Fraser
Authors: R.H. Perrin (2 copies)
R.A. Feik (2 copies)

Materials Research Laboratories

Director/Library

Defence Research Centre

Library

RAN Research Centre

Library

Navy Office

Navy Scientific Adviser
RAN Aircraft Maintenance and Flight Trials Unit (2 copies)
Directorate of Naval Aircraft Engineering
Superintendent, Aircraft Maintenance and Repair

Army Office
- Scientific Adviser - Army
- Engineering Development Establishment, Library
- Royal Military College Library
- US Army Research, Development and Standardisation Group

Air Force Office
- Air Force Scientific Adviser
- Aircraft Research and Development Unit
  - Scientific Flight Group
  - Library
- Technical Division Library
- Director General Aircraft Engineering - Air Force
- Director General Operations Requirements - Air Force
- HQ Operational Command (SMAINTSO)
- HQ Support Command (SLENGO)
- RAAF Academy, Point Cook

Government Aircraft Factories
- Manager
- Library

Department of Aviation
- Library

Statutory and State Authorities and Industry
- Commonwealth Aircraft Corporation, Library
- Hawker de Havilland Aust. Pty Ltd, Bankstown, Library

Universities and Colleges
- Newcastle
  - Department of Electrical and Computer Engineering
    - G.C. Goodwin
    - R.J. Evans

- Sydney
  - Engineering Library
  - Dept. of Aeronautical Engineering

- NSW
  - Physical Sciences Library
  - Dept. of Industrial and Mechanical Engineering

- RMIT
  - Library
  - Department of Civil and Aeronautical Engineering

SPARES (15 copies)
TOTAL (89 copies)

Department of Defence

DOCUMENT CONTROL DATA

1.a. AR No	1.b. Establishment No	2. Document Date	3. Task No
AR-004-475	ARL-AERO-TM-382	APRIL 1986	DST 85/028

4. Title	5. Security a. document	6. No Pages
APPLICATION OF A NOTCH DIGITAL FILTER TO ELIMINATION OF SINUSOIDAL DISTURBANCES FROM HELICOPTER FLIGHT DATA	UNCLASSIFIED b. title U c. abstract U	21 7. No Refs 6

8. Author(s)	9. Downgrading Instructions
R.H. PERRIN, R.A. FEIK	

10. Corporate Author and Address	11. Authority (as appropriate) a.Sponsor b.Security c.Downgrading d.Approval
Aeronautical Research Laboratories P.O. Box 4331, MELBOURNE, VIC. 3001	

12. Secondary Distribution (of this document)

Approved for public release.

Overseas enquirers outside stated limitations should be referred through ASDIS, Defence Information Services Branch, Department of Defence, Campbell Park, CANBERRA ACT 2601

13.a. This document may be ANNOUNCED in catalogues and awareness services available to ...

No Limitations

13.b. Citation for other purposes (ie casual announcement) may be (select) unrestricted(or) as for 13 a.

14. Descriptors	15. COSATI Group
Notch filters Digital filters Sine waves	12010 09020

16. Abstract

Helicopter flight data, especially measurements of linear accelerations and angular velocities, are typically corrupted by sinusoidal deterministic disturbances, that are associated with the rotor frequency and its harmonics. In an effort to eliminate these disturbances without distorting the underlying trends, a new digital notch filter, developed under a research agreement between Aeronautical Research Laboratories (ARL) and the University of Newcastle, has been modified and implemented on the ARL ELXSI 6400 computer. In this memo, the filter is described and the frequency and transient response characteristics are summarised. Practical considerations arising out of application of single and cascaded filters to Sea King flight data are discussed, and the performance of the new filter is compared with that of comparable Butterworth filters.

This paper is to be used to record information which is required by the Establishment for its own use but which will not be added to the DISTIS data base unless specifically requested.

16. Abstract (contd)		
17. Imprint Aeronautical Research Laboratories, Melbourne		
18. Document Series and Number Aerodynamics Technical Memorandum 382	19. Cost Code 51 7770	20. Type of Report and Period Covered
21. Computer Programs Used		
22. Establishment File Ref(s)		

END

12-86

DTIC